

Lesson 7

Second Order Linear Equations a First Encounter

Initializations

```
> restart;
```

7.1 The Harmonic Oscillator.

The differential equation for a harmonic oscillator has the form

$$m \frac{d^2y}{dt^2} + b \frac{dy}{dt} + ky = F_{ext}$$

- m denotes the mass
- b denotes the damping coefficient
- k denotes the spring constant
- F_{ext} denotes an external force
- $y(t)$ denotes the deviation of the mass from the equilibrium position.

Derivational details will be provided in class.

Examples

Example 7.1.1

If $b = 0$ and $F_{ext} = 0$, then

$$m \frac{d^2y}{dt^2} + ky = 0$$

It can easily be seen that $y_1(t) = \sin \sqrt{\frac{k}{m}} t$ and $y_2(t) = \cos \sqrt{\frac{k}{m}} t$ are solutions of this differential equation. Both are oscillating in nature and neither damps out.

Note that for every choice of the constants c_1 and c_2 the function $c_1 y_1(t) + c_2 y_2(t)$ is a solution of the given equation.


Plot

$$s(t) = 5 \sin \sqrt{\frac{k}{m}} t - 3 \cos \sqrt{\frac{k}{m}} t \quad \text{when } k = 2 \text{ and } m = 3.$$

Solution

```
> s:=t->5*sin(sqrt(2/3)*t)-3*cos(sqrt(2/3)*t);  
plot(s(t), t=0..20);
```

$$s := t \rightarrow 5 \sin\left(\sqrt{\frac{2}{3}} t\right) - 3 \cos\left(\sqrt{\frac{2}{3}} t\right)$$



Example 7.1.2

If $m = 1$, $b = 2$, $k = 101$, and $F_{ext} = 0$, then

$$\frac{d^2 y}{dt^2} + 2 \frac{dy}{dt} + 101 y = 0$$

Verify that $y_1(t) = e^{-t} \sin 10 t$ and $y_2(t) = e^{-t} \cos 10 t$ are solutions of this differential equation.

Note 1: In Section 4.3 we will learn how to actually find these solutions.

Note 2: Both y_1 and y_2 represent damped oscillations.

Note 3: This system is said to be under damped.

Note 4: For every choice of the constants c_1 and c_2 the function $c_1 y_1(t) + c_2 y_2(t)$ is a solution of the given equation.

Plot the solution

$$s(t) = 5e^{-t} \sin 10 t - 3e^{-t} \cos 10 t$$

Solution

First code the differential equation and the functions y_1 and y_2 .

```
> deq:=diff(y(t), t$2)+2*diff(y(t), t)+101*y(t)=0;
y1:=t->exp(-t)*sin(10*t);
y2:=t->exp(-t)*cos(10*t);
```

$$deq := \frac{d^2}{dt^2} y(t) + 2 \left(\frac{d}{dt} y(t) \right) + 101 y(t) = 0$$

$$y1 := t \rightarrow e^{-t} \sin(10 t)$$

$$y2 := t \rightarrow e^{-t} \cos(10 t) \quad (2.1.2.1)$$

```
> check1:=eval(deq, y=y1);
check2:=eval(deq, y=y2);
check1 := 0 = 0
check2 := 0 = 0
```

(2.1.2.2)


We conclude that both y_1 and y_2 satisfy the differential equation. Finally we plot the solution

$$s(t) = 5e^{-t} \sin 10 t - 3e^{-t} \cos 10 t$$

```
> s:=unapply(5*y1(t)-3*y2(t), t);
```

```
plot(s(t), t=0..3);
```

```
s := t -> 5 e^{-t} sin(10 t) - 3 e^{-t} cos(10 t)
```



Example 7.1.3

If $m = 1$, $b = 3$, $k = 2$, and $F_{ext} = 0$, then

$$\frac{d^2 y}{dt^2} + 3 \frac{dy}{dt} + 2y = 0$$

Verify that $y_1(t) = e^{-t}$ and $y_2(t) = e^{-2t}$ are solutions of this differential equation.

Note 1: In Section 4.2 we will learn how to actually find these solutions.

Note 2: Neither of these solutions is oscillatory.

Note 4: This system is said to be over damped.

Note 4: For every choice of the constants c_1 and c_2 the function $c_1 y_1(t) + c_2 y_2(t)$ is a solution of the given equation.

Plot the solution

$$s(t) = 5e^{-t} - 3e^{-2t}$$

Solution

First code the differential equation and the functions y_1 and y_2 .

```
> deq:=diff(y(t), t$2)+3*diff(y(t), t)+2*y(t)=0;  
y1:=t->exp(-t);  
y2:=t->exp(-2*t);
```

$$deq := \frac{d^2}{dt^2} y(t) + 3 \left(\frac{d}{dt} y(t) \right) + 2 y(t) = 0$$

$$y1 := t \rightarrow e^{-t}$$

$$y2 := t \rightarrow e^{-2t}$$

(2.1.3.1)


```
> check1:=eval(deq, y=y1);  
check2:=eval(deq, y=y2);  
check1 := 0 = 0  
check2 := 0 = 0
```

(2.1.3.2)

We conclude that both y_1 and y_2 satisfy the differential equation. Finally we plot the solution

$$s(t) = 5e^{-t} - 3e^{-2t}$$

```
> s:=unapply(5*y1(t)-3*y2(t), t);  
plot(s(t), t=0..10);
```



Example 7.1.4

If $m = 1$, $b = 4$, $k = 4$, and $F_{ext} = 0$, then

$$\frac{d^2 y}{dt^2} + 4 \frac{dy}{dt} + 4 y = 0$$

Verify that $y_1(t) = e^{-2t}$ and $y_2(t) = t e^{-2t}$ are solutions of this differential equation.

Note 1: In Section 4.2 we will learn how to actually find these solutions.

Note 2: Neither of these solutions is oscillatory.

Note 4: This system is said to be critically damped.

Note 4: For every choice of the constants c_1 and c_2 the function $c_1 y_1(t) + c_2 y_2(t)$ is a solution of the given equation.

Plot the solution

$$s(t) = 5e^{-2t} - 3t e^{-2t}$$

Solution

First code the differential equation and the functions y_1 and y_2 .

```
> deq:=diff(y(t), t$2)+4*diff(y(t), t)+4*y(t)=0;
y1:=t->exp(-2*t);
y2:=t->t*exp(-2*t);
```

$$deq := \frac{d^2}{dt^2} y(t) + 4 \left(\frac{d}{dt} y(t) \right) + 4 y(t) = 0$$

$$y1 := t \rightarrow e^{-2t}$$

$$y2 := t \rightarrow t e^{-2t} \tag{2.1.4.1}$$


```
> check1:=eval(deq, y=y1);
check2:=eval(deq, y=y2);
check1 := 0 = 0
check2 := 0 = 0
```

$$\tag{2.1.4.2}$$

We conclude that both y_1 and y_2 satisfy the differential equation. Finally we plot the solution

$$s(t) = 5e^{-2t} - 3t e^{-2t}$$

```
> s:=unapply(5*y1(t)-3*y2(t), t);
plot(s(t), t=0..2);
s := t -> 5 e^{-2t} - 3 t e^{-2t}
```



Example 7.1.5

If a mass-spring system is driven by an external force that is sinusoidal at the angular frequency ω , then eventually it will respond in "sync" with the driver and oscillate at the same frequency.

Find the synchronous response of the mass-spring oscillator with $m = 1$, $b = 1$, $k = 25$ to the force $\sin \Omega t$. Investigate the amplitude of the response as a function of Ω .

Solution

We will seek a solution of the differential equation

$$y'' + y' + 25 y = \sin \Omega t$$

that is of the form

$$y = A \cos \Omega t + B \sin \Omega t$$

Code the differential equation and the form of the sought after solution.

```
> deq:=diff(y(t), t$2)+diff(y(t), t)+25*y(t)=sin(Omega*t);
ytry:=t->A*cos(Omega*t)+B*sin(Omega*t);
```

$$deq := \frac{d^2}{dt^2} y(t) + \frac{d}{dt} y(t) + 25 y(t) = \sin(\Omega t)$$

$$ytry := t \rightarrow A \cos(\Omega t) + B \sin(\Omega t) \quad (2.1.5.1)$$

Substitute the coded solution into the differential equation. Then find the constants A and B by comparing the coefficients of $\cos \Omega t$ and $\sin \Omega t$ on both sides.

```
> e1:=eval(deq, y=ytry);
```

$$e1 := -A \cos(\Omega t) \Omega^2 - B \sin(\Omega t) \Omega^2 - A \sin(\Omega t) \Omega + B \cos(\Omega t) \Omega + 25 A \cos(\Omega t) + 25 B \sin(\Omega t) = \sin(\Omega t) \quad (2.1.5.2)$$

```
> pars:=solve(identity(e1, t), {A, B});
```

$$pars := \left\{ A = -\frac{\Omega}{-49 \Omega^2 + \Omega^4 + 625}, B = -\frac{\Omega^2 - 25}{-49 \Omega^2 + \Omega^4 + 625} \right\} \quad (2.1.5.3)$$

Hence, the synchronous solution is given by

```
> ys:=unapply(eval(ytry(t), pars), t);
```

$$ys := t \rightarrow -\frac{\Omega \cos(\Omega t)}{-49 \Omega^2 + \Omega^4 + 625} - \frac{(\Omega^2 - 25) \sin(\Omega t)}{-49 \Omega^2 + \Omega^4 + 625} \quad (2.1.5.4)$$

To find the amplitude of this solution as a function of Ω we use the trigonometric relationship

$$\sin(\alpha + \beta) = \sin \alpha \cos \beta + \cos \alpha \sin \beta$$

and write

$$\begin{aligned} A \cos \Omega t + B \sin \Omega t &= \sqrt{A^2 + B^2} \left(\frac{A}{\sqrt{A^2 + B^2}} \cos \Omega t + \frac{B}{\sqrt{A^2 + B^2}} \sin \Omega t \right) \\ &= \sqrt{A^2 + B^2} \sin(\Omega t + \varphi) \end{aligned}$$

where $\varphi = \arctan(A, B)$.


Hence the amplitude of the synchronous solution equals $\sqrt{A^2 + B^2}$.

```
> Ampl:=simplify(eval(sqrt(A^2+B^2), pars));
```

$$Ampl := \sqrt{\frac{1}{-49 \Omega^2 + \Omega^4 + 625}} \quad (2.1.5.5)$$

We can now plot this amplitude as a function of Ω .

```
> plot(Ampl, Omega=0..10);
```



Clearly, there is a spike of the amplitude near $\Omega = 5$. This phenomenon is known as resonance, which will be discussed in section 4.10. The actual resonance frequency can be obtained by solving the equation

$$\frac{d}{d\Omega} Ampl(\Omega) = 0$$

for Ω .

```
> eq:=diff(Ampl, Omega)=0;
```

$$eq := -\frac{1}{2} \frac{-98 \Omega + 4 \Omega^3}{\sqrt{\frac{1}{-49 \Omega^2 + \Omega^4 + 625}} (-49 \Omega^2 + \Omega^4 + 625)^2} = 0 \quad (2.1.5.6)$$

```
> val_Omega:=solve(eq, Omega);
```

$$val_Omega := 0, \frac{7}{2} \sqrt{2}, -\frac{7}{2} \sqrt{2} \quad (2.1.5.7)$$

We conclude that the (circular) resonance frequency equals $\Omega = \frac{7}{2} \sqrt{2}$ which is close to 5, but not equal to 5.

```
> evalf(val_Omega[2]);  
4.949747467
```

(2.1.5.8)